微積分 (多変数)

Student パッケージには、授業で役立つたくさんの機能が含まれています。Maple 17 では、いくつかの新機能が追加されました。最も重要な新機能として、MultivariateCalculus サブパッケージに Line および Plane オブジェクトが導入されています。ここでは、これらのオブジェクトについて説明します。

Line および Plane は、簡単な高校レベルの (アフィン) 2 次元または 3 次元幾何学を教えるためのオブジェクトです。オブジェクトはさまざまな方法で定義できます。たとえば、3 次元空間の Line は以下のように定義できます。

- 2 つの点を含みます。
- 1 つの点と方向を含みます。
- 2 つの方程式の解となります。
- パラメトリック表現を持ちます。
- 1 つの点を含み、平面に直交しています。または、
- 2 つの平面に含まれています。

オブジェクトが作成されたら、別のオブジェクトまでの距離、相対位置、複数オブジェクト間の交点を求めることができます。

多くのアプリケーションで、オブジェクトが表す線と平面を完全に求めることができます。ただし、Maple ではオブジェクトを定義するのに使用されたパラメータの座標で生じた任意の代数式をサポートしています。たとえば、点 $[2, 3, a]$ と $[1, a^2 + b, 3]$ を通る線を扱うことができます。

例

> $with(Student\text{:-}MultivariateCalculus):$

最初の線には点 $[1, 0, 2]$ と方向 $\langle 2, -2, 1\rangle$ が含まれ、2 つ目の線には点 $[5, -3, 1]$ と $[3, -3, 6]$ が含まれる 2 つの線を作成します。Line および Plane オブジェクトは、lists を点として、Vectors を方向として認識します。

> $l1 := Line([1, 0, 2], \langle 2, -2, 1\rangle)$

$$l1 := <<Line\ 1>> \tag{1}$$

> $l2 := Line([5, -3, 1], [3, -3, 6])$

$$l2 := <<Line\ 2>> \tag{2}$$

$l1$ が $l2$ と交差するかどうかを確認してみましょう。

> $Intersects(l1, l2)$

$$\text{true} \tag{3}$$

> $pt := GetIntersection(l1, l2)$

$$pt := \left[4, -3, \frac{7}{2}\right] \tag{4}$$

交点は点です。

> $Contains(l1, pt)$

$$\text{true} \tag{5}$$

> $Contains(l2, pt)$

$$\text{true} \tag{6}$$

両方の線の方程式 (それぞれの線で 2 つの式) を取得して *pt* を求め、同時に解くこともできます。デフォルトの座標変数は *x*、*y*、および *z* ですが、線または平面を作成するときに別の変数を選択することができます。

> $eqns1 := GetRepresentation(l1, 'form = equations')$

$$eqns1 := \left\{x + y = 1, -\frac{1}{2}\,x + z = \frac{3}{2}\right\} \tag{7}$$

> $eqns2 := GetRepresentation(l2, 'form = equations')$

$$eqns2 := \left\{y = -3, \frac{5}{2}\,x + z = \frac{27}{2}\right\} \tag{8}$$

> $solve(eqns1 \cup eqns2)$

$$\left\{x = 4, y = -3, z = \frac{7}{2}\right\} \tag{9}$$

GetRepresentation コマンドを使用すると、線をさまざまな表現で取得できます。

> $GetRepresentation(l1)$

$$t.\begin{bmatrix} 2 \\ -2 \\ 1 \end{bmatrix} + \begin{bmatrix} 1 \\ 0 \\ 2 \end{bmatrix} \tag{10}$$

> $GetRepresentation(l1, 'form = combined\_vector')$

$$\begin{bmatrix} 1 + 2\,t \\ -2\,t \\ 2 + t \end{bmatrix} \tag{11}$$

> $GetRepresentation(l1, 'form = parametric')$

$$[x = 1 + 2\,t, y = -2\,t, z = 2 + t] \tag{12}$$

> $GetRepresentation(l1, 'form = symmetric')$

$$\frac{x}{2} - \frac{1}{2} = -\frac{y}{2} = z - 2 \tag{13}$$

l1 に平行な 3 つ目の線を作成します。

> $l3 := Line([0, 0, 0], [2, -2, 1])$

$$l3 := << Line\ 3 >> \tag{14}$$

> $AreParallel(l1, l3)$

$$\text{true} \tag{15}$$

l2 を基準とした *l3* の相対位置はどこでしょうか？

> $AreParallel(l2, l3)$

$$\text{false} \tag{16}$$

> $Intersects(l2, l3)$

$$\text{false} \tag{17}$$

> $AreSkew(l2, l3)$

$$\text{true} \tag{18}$$

Distance コマンドを使用すると、2 つの線のあいだの (ユークリッド) 距離を計算できます。交差する線は距離が 0 です。

> $Distance(l1, l2)$

$$0 \tag{19}$$

> $Distance(l1, l3)$

$$\frac{1}{3}\sqrt{29} \tag{20}$$

> $Distance(l2, l3)$

$$\frac{9}{65}\sqrt{65} \tag{21}$$

GetPlot コマンドは、線を可視化して表示します。

> $GetPlot(l1)$



The line $[1+2t, -2t, 2+t]$. A point on the line. The direction vector.

可視化を組み合わせるには、plots:-display を使用します。すべての可視化機能をオンにすると見にくくなるため、ここではいくつかをオフにします。

> $plots{:}\text{-}display(seq(GetPlot(line, 'showvector = false', 'showpoint = false'), line \in [l1, l2,$

$l3$]), '*caption*' = "Three lines")



Three lines

$l1$ と $l2$ を含む平面を考えてみましょう。

> $p1 := Plane(l1, l2)$

$$p1 := << Plane\ 1 >> \tag{22}$$

$l3$ と $p1$ の相対位置はどこでしょうか？

> $Intersects(p1, l3)$

$$\text{false} \tag{23}$$

> $AreParallel(p1, l3)$

$$\text{true} \tag{24}$$

> $Distance(p1, l3)$

$$\frac{9}{65}\sqrt{65} \tag{25}$$

$p1$ と $l3$ の距離は $l2$ と $l3$ の距離と同じです。$l3$ が $p1$ に平行な場合 ($p1$ には $l2$ が含まれるが $l2$ は $l3$ に平行ではない)、これは常に当てはまります。

$l1$ と $l2$ は交差しているため、$l2$ も $l1$ と $l3$ を含む平面と交差します。

> $p2 := Plane(l1, l3)$

$$p2 := << Plane\ 2 >> \tag{26}$$

> $Intersects(p2, l2)$

$$\text{true} \tag{27}$$

ここで、いくつかの線を考えてみます。値 a と b に対して、$l4$ を点 $[1, 2, -2]$ と方向 $\langle a, b, 1\rangle$ を含む線とします。

> $l4 := Line([1, 2, -2], \langle a, b, 1\rangle)$

$$l4 := <<Line\ 4>> \tag{28}$$

> $Intersects(l4, l2)$

$$\text{false} \tag{29}$$

> $Distance(l4, l2)$

$$\frac{|25\,a + 26\,b + 10|}{\sqrt{29\,|b|^2 + |2 + 5\,a|^2}} \tag{30}$$

> $Distance(l4, l3)$

$$\frac{|2\,a + 5\,b + 6|}{\sqrt{|b + 2|^2 + |-2 + a|^2 + |2\,b + 2\,a|^2}} \tag{31}$$

両方の距離の分子がゼロになる a と b の値を求められれば、$l2$ と $l3$ の両方と交差する線を得ることができます。

> $solve(\{numer(\mathbf{(30)}) = 0, numer(\mathbf{(31)}) = 0\})$

$$\left\{a = \frac{106}{73}, b = -\frac{130}{73}\right\} \tag{32}$$

そして、$l5$ を a と b の値を持つ線とします。

> $l5 := eval(l4, \mathbf{(32)})$

$$l5 := <<Line\ 5>> \tag{33}$$

> $GetRepresentation(l5)$

$$t.\begin{bmatrix} \frac{106}{73} \\ -\frac{130}{73} \\ 1 \end{bmatrix} + \begin{bmatrix} 1 \\ 2 \\ -2 \end{bmatrix} \tag{34}$$

> $GetIntersection(l2, l5)$

$$\left[\frac{66}{13}, -3, \frac{21}{26}\right] \tag{35}$$

> $GetIntersection(l3, l5)$

$$\left[\frac{57}{4}, -\frac{57}{4}, \frac{57}{8}\right] \tag{36}$$

その他の例については、MultivariateCalculus の例題ワークシートを参照してください。

参照

AreSkew、GetIntersection、GetRepresentation、Intersects、Line、Plane